#-------------------------------------------------
#
# Project created by QtCreator 2018-10-08T11:03:05
#
#-------------------------------------------------
QT                     -= core gui

TARGET                  = log4x
TEMPLATE                = lib
CONFIG                 += plugin
DEFINES                += LOG4X_EXPORTS

win32{
contains(QMAKE_COMPILER, 'msvc') {
COMPILER                = \"MSVC-$${QMAKE_MSC_FULL_VER}\"
QMAKE_CXXFLAGS_WARN_ON += -wd4819
}else{

COMPILER                = \"MinGW-$${QMAKE_GCC_MAJOR_VERSION}.$${QMAKE_GCC_MINOR_VERSION}.$${QMAKE_GCC_PATCH_VERSION}\"
QMAKE_CXXFLAGS         += -Wno-unused-parameter -Wno-missing-field-initializers
}
DEFINES                += VERSION_COMPILER=$${COMPILER}
DEFINES                -= UNICODE
message($${COMPILER})
}

!debug_and_release|build_pass {
    CONFIG(debug, debug|release) {
        TARGET          = ../../lib/$$member(TARGET, 0)d
    }
}

head.files    = \
    ../src/log4x.h \
    ../src/log4x-c.h \
    ../src/echo/log4xecho.h

SOURCES      += \
    ../src/sys.cpp \
    ../src/log4x.cpp \
    ../src/log4x-c.cpp \
    ../src/concurrency.cpp \
    ../src/thread.cpp \
    ../src/echo/log4xecho.cpp


HEADERS      += \
    ../src/file.h \
    ../src/log4x.h \
    ../src/sys.h \
    ../src/concurrency.h \
    ../src/stream.h \
    ../src/thread.h

#target.path  = ../lib
#INSTALLS    += target

unix{
HEADERS      += ../src/usys.h
SOURCES      += ../src/usys.cpp
}

win32{
HEADERS      += ../src/wsys.h
SOURCES      += ../src/wsys.cpp
LIBS         += -lshlwapi
}

INCLUDEPATH  += \
    ../ \
    ../src

RC_FILE       = ../src/log4x.rc
